NAG Library Routine Document

F07WXF (ZTFTRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07WXF (ZTFTRI) computes the inverse of a complex triangular matrix stored in Rectangular Full Packed (RFP) format. The RFP storage format is described in Section 3.3.3 in the F07 Chapter Introduction.

2 Specification

```
SUBROUTINE F07WXF (TRANSR, UPLO, DIAG, N, A, INFO)

INTEGER                 N, INFO
COMPLEX (KIND=nag_wp) A(N*(N+1)/2)
CHARACTER(1)            TRANSR, UPLO, DIAG
```

The routine may be called by its LAPACK name ***ztftri***.

3 Description

F07WXF (ZTFTRI) forms the inverse of a complex triangular matrix A, stored using RFP format. Note that the inverse of an upper (lower) triangular matrix is also upper (lower) triangular.

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

Gustavson F G, Waśniewski J, Dongarra J J and Langou J (2010) Rectangular full packed format for Cholesky's algorithm: factorization, solution, and inversion *ACM Trans. Math. Software* **37, 2**

5 Parameters

1: TRANSR – CHARACTER(1) *Input*

On entry: specifies whether the normal RFP representation of A or its conjugate transpose is stored.

TRANSR = 'N'
The matrix A is stored in normal RFP format.

TRANSR = 'C'
The conjugate transpose of the RFP representation of the matrix A is stored.

Constraint: TRANSR = 'N' or 'C'.

2: UPLO – CHARACTER(1) *Input*

On entry: specifies whether A is upper or lower triangular.

UPLO = 'U'
A is upper triangular.

UPLO = 'L'
A is lower triangular.

Constraint: UPLO = 'U' or 'L'.

3: DIAG – CHARACTER(1) *Input*

On entry: indicates whether A is a nonunit or unit triangular matrix.

DIAG = 'N'
A is a nonunit triangular matrix.

DIAG = 'U'
A is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.

Constraint: DIAG = 'N' or 'U'.

4: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

5: A(N $\times$ (N + 1)/2) – COMPLEX (KIND=nag_wp) array *Input/Output*

On entry: the n by n triangular matrix A, stored in RFP format.

On exit: A is overwritten by A^{-1}, in the same storage format as A.

6: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, $a(i,i)$ is exactly zero; A is singular and its inverse cannot be computed.

7 Accuracy

The computed inverse X satisfies

$$|XA - I| \leq c(n)\epsilon|X||A|,$$

where $c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

Note that a similar bound for $|AX - I|$ cannot be guaranteed, although it is almost always satisfied.

The computed inverse satisfies the forward error bound

$$\left|X - A^{-1}\right| \leq c(n)\epsilon\left|A^{-1}\right||A||X|.$$

See Du Croz and Higham (1992).

8 Further Comments

The total number of real floating point operations is approximately $\frac{4}{3}n^3$.

The real analogue of this routine is F07WKF (DTFTRI).

9 Example

This example computes the inverse of the matrix A, where

$$A = \begin{pmatrix} 4.78 + 4.56i & 0.00 + 0.00i & 0.00 + 0.00i & 0.00 + 0.00i \\ 2.00 - 0.30i & -4.11 + 1.25i & 0.00 + 0.00i & 0.00 + 0.00i \\ 2.89 - 1.34i & 2.36 - 4.25i & 4.15 + 0.80i & 0.00 + 0.00i \\ -1.89 + 1.15i & 0.04 - 3.69i & -0.02 + 0.46i & 0.33 - 0.26i \end{pmatrix}$$

and is stored using RFP format.

9.1 Program Text

```
    Program f07wxfe

!     F07WXF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, ztftri, ztfttr
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: ifail, info, ldf, lena, n
      Character (1)                    :: diag, transr, uplo
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:), f(:,:)
      Character (1)                    :: clabs(1), rlabs(1)
!     .. Executable Statements ..
      Write (nout,*) 'F07WXF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, uplo, transr, diag

      lena = n*(n+1)/2
      ldf = n
      Allocate (a(1:lena),f(ldf,n))

!     Read A from data file
      Read (nin,*) a(1:lena)

!     Compute inverse of A
!     The NAG name equivalent of ztftri is f07wxf
      Call ztftri(transr,uplo,diag,n,a,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       Convert and print inverse
!       The NAG name equivalent of ztfttr is f01vhf
        Call ztfttr(transr,uplo,n,a,f,ldf,info)
        ifail = 0

        Call x04dbf(uplo,'Nonunit',n,n,f,ldf,'Bracketed','F7.4','Inverse', &
          'Integer',rlabs,'Integer',clabs,80,0,ifail)
      Else
        Write (nout,*) 'A is singular'
      End If

    End Program f07wxfe
```

9.2 Program Data

```
F07WXF Example Program Data
  4 'L' 'N' 'N'  : n, uplo, transr, diag
 ( 4.15,-0.80)
 ( 4.78, 4.56)
 ( 2.00,-0.30)
 ( 2.89,-1.34)
 (-1.89, 1.15)

 (-0.02,-0.46)
 ( 0.33, 0.26)
 (-4.11, 1.25)
 ( 2.36,-4.25)
 ( 0.04,-3.69)   : A in RFP storage
```

9.3 Program Results

```
 F07WXF Example Program Results

 Inverse
                     1                 2                 3                 4
 1  ( 0.1095,-0.1045)
 2  ( 0.0582,-0.0411) (-0.2227,-0.0677)
 3  ( 0.0032, 0.1905) ( 0.1538,-0.2192) ( 0.2323,-0.0448)
 4  ( 0.7602, 0.2814) ( 1.6184,-1.4346) ( 0.1289,-0.2250) ( 1.8697, 1.4731)
```